

Højereordens Funktioner.

Motivation: mapping.
Motivation: parameterombytning.
Sektioner af binære operatører.
Currying.
Filtrering.
Sammensætning.
Produkt.
Reducering og akkumulering.

PS3 - Højereordens funktioner

© Kurt Nørmark, Aalborg Universitet.

10/3/96 s. 1

Noter

Som omtalt tidligere, er højereordens funktioner et nøgleområde inden for funktionsorienteret programmering.

Alle scheme funktioner i dette afsnit findes på filen

`/user/normark/public/ps4/ho-fn.scm`.

Funktionerne kan også nåes via denne lektion's side på WWW.

En funktion som ombytter sine parametre.

En funktion

$rev: ((X \times Y) \rightarrow Z) \rightarrow (Y \times X) \rightarrow Z$
 $rev(f(x,y)) = f(y,x)$

```
(define (rev f)
  (lambda (x y)
    (f y x)))
```

Andre "tilsvarende højereordens funktioner":

$derive: (R \rightarrow R) \times R \rightarrow (R \rightarrow R)$
 $inverse: (X \rightarrow Y) \rightarrow (Y \rightarrow X)$

```
1> (- 2 3)
-1
2> ((rev -) 2 3)
1
3> (member 3
      (list 1 2 3))
(3)
4> ((rev member)
      (list 1 2 3) 3)
(3)
5> (cons 5 6)
(5 . 6)
6> ((rev cons) 5 6)
(6 . 5)
```

PS3 - Højereordens funktioner

© Kurt Nørmark, Aalborg Universitet.

10/3/96 s. 2

Noter

Vi har ved tidligere forelæsninger set på funktioner som afleder en funktion med hensyn til en variable. I forbindelse med introduktionen af Lisp studerende vi symbolsk differentiering. Senest har vi set på funktionen `derive`, som jo returnerede den numerisk differentierede funktion af en funktion. Vi minder her om, at `derive` var en højereordens funktion ligesom `rev` ovenfor.

Inverse er funktionen som returnerer den inverse funktion f^{-1} af en funktion f . Denne opfylder:

$f^{-1}(f(x)) = x$ for alle x i X .

Der er naturligvis vanskeligheder og udfordringer i at definere funktionen `inverse`. For hvilke funktioner skal den gælde? Hvordan bærer vi os ad med at bestemme den?

Opgave: En simple idé kunne være at tabellægge funktioner og deres inverse, og at lade funktionen `inverse` benytte sig af dette. Demonstrer denne ide i praksis i Scheme.

Mapping.

Det er let at skrive en funktion, som “mapper” en bestemt funktion på en liste:

```
(define (map-square lst)
  (if (null? lst)
      '()
      (cons (square (car lst))
            (map-square (cdr lst)))))
```

Denne kan let **generaliseres** til at mappe en vilkårlig funktion på listen.

```
(define (map f lst)
  (if (null? lst)
      '()
      (cons (f (car lst))
            (map f (cdr lst)))))
```

```
> (map square
    (list 1 2 3 4 5))
(1 4 9 16 25)
```

Det ville være nyttigt og kunne håndtere funktionen `(map f)` som et førsteklases objekt.

PS3 - Højereordens funktioner

Noter

Mapping hører så absolut til blandt de klassiske højereordens funktioner. Endvidere er mapping anvendelig i mange praktiske sammenhænge.

Mapning af en funktion på en liste foregår ved at anvende funktionen på hvert af listens elementer, hvorved der fremkommer og opsamles nye elementer, der igen danner en liste. Mapning af en funktion på en liste L giver således en ny liste af samme længde som L.

En praktisk bemærkning: Husk at den tomme liste i Scheme skal skrives som `()`. Det er altså ikke nok blot at skrive `()`.

Generering af en map-funktion.

```
(define (map f)
  (lambda (lst)
    (if (null? lst)
        '()
        (cons (f (car lst))
              ((map f) (cdr lst)))))))
```

```
1> (map square)
#<Function>

2> (let
    ((f (map square)))
  (f (list 1 2 3 4 5)))
(1 4 9 16 25)

3> ((map square)
    (list 1 2 3 4 5))
(1 4 9 16 25)
```

- Signatur af map:

$$(X \rightarrow Y) \rightarrow (X^* \rightarrow Y^*)$$

- Signatur af forrige map:

$$(X \rightarrow Y) \times X^* \rightarrow Y^*.$$

- Funktionen (map f), for en vilkårlig funktion f , er en byggeklod i funktions-orienteret programmering, der kan bruges som parameter, f.eks i andre mapninger:

```
> ((map (map square)) '((1 2 3) (7 8 9)))
((1 4 9) (49 64 81))
```

PS3 - Højereordens funktioner

Noter

Man kan opfatte ovenstående version af map som en funktionsgenerator. Det tager en funktion som input (som parameter) og giver en funktion som output (som resultat, returværdi).

Undertiden kalder man højereordensfunktioner for funktionale.


Hvis man studerer ovenstående version af map nærmere vil man måske få øje på et problem. Det problem vi har i tankerne er

(map f)

kaldet i den sidste linie i funktionen. Her kaldes map rekursivt, hvorved der genereres og returneres *et nyt funktionsobjekt* fra map. Dette nye funktionsobjekt er lige så god som den oprindelige (map f) funktion, der blev skabt ifm. anvendelsen af map. Men der er ingen grund til at lave nye funktionsobjekter gentagne gange. Det kræver både tid, og tager plads, som blot skal genindrives igen. Nedenunder har vi arrangeret tingene således, at denne re-generering ikke finder sted. I stedet for er local-map nu et frit navn i lambda-udtrykket, som er bundet i en letrec form mellem map niveauet og lambdaudtrykket:

```
(define (map f)
  (letrec
    ((local-map
      (lambda (lst)
        (if (null? lst)
            '()
            (cons (f (car lst))
                  (local-map (cdr lst)))))))
    local-map))
```

Eksempel: (map map).



`(map map): (A -> B)* -> (A* -> B)*`

Fortolkning: Funktionen `(map map)` afbilder en liste af funktioner over i en liste af *mapping funktioner*.

```
1> ((map map) (list power2 square))           ~ (power2-map square-map)
    (#<Function> #<Function>)                 Vi anvender successivt power2-map
                                                og square-map på listen (1 .. 8)

2> (let ((f-map-list ((map map) (list power2 square))))
      ((map (lambda(f) (f (interval 1 8)))) f-map-list))

((2 4 8 16 32 64 128 256) (1 4 9 16 25 36 49 64))
```

PS3 - Højereordens funktioner

© Kurt Nørmark, Aalborg Universitet.

10/3/96 s. 5

Noter

Ovenstående er et lidt mere kompliceret eksempel. Eksemplet er spændende idet vi anvender `map` på sig selv. En sådan form for "selvanvendelse" virker ofte dragende, og det forekommer i mange forskellige sammenhænge. (Tænk for eksempel på en compiler til et programmeringssprog: skriv compileren i sproget selv, og lad dernæst compileren oversætte sig selv).

Både den yderste og inderste `map` har den signatur, som vi studerede på forrige slides. Vi anvender symbolerne X og Y i den yderste, og A og B i den inderste. Den yderste `map` skal anvendes på en funktion $X \rightarrow Y$. Den inderste `map` opfylder dette krav: $X = A \rightarrow B$ og $Y = (A^* \rightarrow B^*)$. Billedmængden af den yderste `map` er $(X^* \rightarrow Y^*)$. Når X substitueres med $A \rightarrow B$ og Y med $(A^* \rightarrow B^*)$ får vi ovenstående signatur for `(map map)`.

Det ses altså, at `(map map)` tager en liste af funktioner, og at den returnerer en liste af *mapping-funktioner* (hvormed vi mener funktioner mapper en bestemt funktion hen over en liste). I det ovenstående kalder vi konkret de to resulterende mapping funktioner for `power2-map` og `square-map`.

Bemærk, at i en praktisk sammenhæng kan det være upraktisk eller måske ulovligt at redefinere `map`, som er pre-defineret i Scheme. Det anbefales, at kalde `map`, som blev defineret på forrige slide, for noget andet end `map`. I filen med højereordensfunktioner har jeg kaldt den `map2`.

Her følger definitionerne på funktionerne `power2` og `interval`, som bruges ovenfor i eksemplet.

Først `interval`, som genererer en *liste* af tal fra og med x til og med y .

```
(define (interval x y)
  (if (> x y)
      '()
      (cons x (interval (+ x 1) y))))
```

Udtrykket `(power2 x)` returnerer 2^x :

```
(define (power2 x)
  (if (= x 0) 1 (* 2 (power2 (- x 1)))))
```

Currying

Currying er en systematik som transformerer en funktion af flere parametre til en funktion af én parameter:

$f: A \times B \rightarrow C$ *ikke curried.*

$f: A \rightarrow (B \rightarrow C)$ *curried.*

Eksempler:

- $+ 5 6$ fortolkes som $[+ 5] 6$
[+ 5] er 'en funktion som adderer 5 til sin parameter'.
- $(+ 5 6)$ svarer i curried version til $((\text{lambda}(x) (+ 5 x)) 6)$

Nogle funktions-orienterede sprog understøtter kun curried funktioner.

Positiv side: Med notationsmæssig behændighed opnåes, at prefixes af et udtryk er meningsfuldt, nemlig en funktion.

Negativ side: Asymmetri i parametrene.

Negativ side: Mister muligheden for at betragte parameter-listen som første klasses objekt (en tupel).

PS3 - Højereordens funktioner

Noter

På de første slides i dette kapitel startede vi med at se på ikke curried og curried udgave af mapping.

Currying *generaliserer* til funktioner af mere end to parametre:

$f: A_1 \times A_2 \times A_3 \dots \times A_n \rightarrow B$ ikke-curried.

$(f a_1 a_2 a_3 \dots a_n)$

$f: A_1 \rightarrow (A_2 \rightarrow (A_3 \rightarrow \dots (A_n \rightarrow B) \dots))$ curried.

$(((((f a_1) a_2) a_3) \dots) a_n)$

Generelt er anvendelse af curried funktioner notationsmæssig tung i Lisp på grund af de eksplicite parenteser. Man får jo udtryk på formen $(((((f a_1) a_2) a_3) \dots) a_n)$. Currying af binære funktioner giver sig udslag i $((f a) b)$.

Et funktionsorienteret sprog uden Lisp parenteser er notationsmæssig overlegen her. I et sådant sprog kan vi let opsætte konventioner så $(((((f a_1) a_2) a_3) \dots) a_n)$ blot skrives som

$f a_1 a_2 a_3 \dots a_n$

Konventionen er *venstreassociativitet*, og at alle funktioner er unære (altså af én parameter).

Bemærk at signaturen

$f: A_1 \rightarrow (A_2 \rightarrow (A_3 \rightarrow \dots (A_n \rightarrow B) \dots))$

er associerer til højre.

Generering af curried funktioner.

```
(define (curry f)
  ; f: A x B -> C er en binær,
  ; ikke curried funktion
  (lambda(x)
    (lambda(y)
      (f x y))))
```

```
(define (uncurry f)
  ; f: A -> (B -> C) er en
  ; curried funktion
  (lambda (x y)
    ((f x) y)))
```

```
1> (((curry +) 5) 6)
11
```

```
2> ((uncurry (curry +)) 5 6)
11
```

```
3> ((map square) (list 5 6 7))
(25 36 49)
```

```
4> ((uncurry map) square (list 5 6 7))
(25 36 49)
```

PS3 - Højereordens funktioner

© Kurt Nørmark, Aalborg Universitet.

10/3/96 s. 7

Noter

Signatur for curry:

$$(A \times B \rightarrow C) \rightarrow (A \rightarrow (B \rightarrow C))$$

Signatur for uncurry:

$$(A \rightarrow (B \rightarrow C)) \rightarrow (A \times B \rightarrow C)$$

Opgave: Fortolk følgende udtryk.

```
(let ((cm (curry -))) (map cm (list 1 2 3 4 5))) [1]
```

Som det ses, antager vi at map her er uncurried.

Hvis resultatet af udtrykket [1] betegnes med res, hvad giver da følgende udtryk?

```
(map (lambda(x) (x 5)) res)
```

Løsningen findes på notearket til sidste slide i dette kapitel.

Sektioner af binære operatører.

Binær infix operation	$x \text{ op } y$	$(\text{op } x \ y)$
Venstre sektion	$[x \ \text{op}] \ y$	$((\text{lambda } (z) (\text{op } x \ z)) \ y)$
Højre sektion	$x \ [\text{op} \ y]$	$((\text{lambda } (z) (\text{op } z \ y)) \ x)$

```
1> (+ 1 5)
6
```

```
2> (define (1+ x)
      (+ 1 x))
```

```
3> (1+ 5)
6
```

Generering af sektioner:

```
(define (left-section OP x) ; fixing x
      (lambda (y)
        (OP x y)))
```

```
(define (right-section OP y) ; fixing y
      (lambda (x)
        (OP x y)))
```

```
4> (define lig-med-nul
      (left-section = 0))
lig-med-nul
5> (lig-med-nul 5)
#f
```

```
6> (define unit-list
      (right-section cons '()))
unit-list
7> (unit-list 5)
(5)
```

PS3 - Højereordens funktioner

Noter

Venstresektionering låser den venstre parameter af en binær operator. Tilsvarende låser højresektionering den højre parameter.

Venstresektionering af en binær operator op svarer nøje til currying af op . Højresektionering af op svarer til currying af (rev op) , hvor rev er funktionen som bytter parametrene af en binær funktion om.

Når vi anvender notationen $[x \ \text{op}]$ og $[\text{op} \ y]$ ovenfor ser vi, at venstre sektionering skal fortolkes som prefix notation (operator før operand), og højresektionering skal fortolkes som postfix notation (operand før operator).

Som illustreret ovenfor, kan venstresektionering give os en tilnærmet infix notation, hvis vi er smarte med navngivning af den resulterende funktion. Eksempelvis:

$$(1- \ 3) = (- \ 1 \ 3) = 1 - 3.$$

Der kan i nogle Scheme systemer være leksikalske problemer ved navnet '1-', som vi benyttede ovenfor, idet det starter med et tal. Endvidere kan man forvirre sig selv og andre ved at bruge notationen. Så måske skal vi glemme det igen...

Højresektionerede funktioner kan navngives tilsvarende:

$$(-1 \ 3) = (- \ 3 \ 1) = 3 - 1.$$

Bemærk, at højre sektionering ikke notationsmæssig er så elegant som venstre sektionering, idet vi jo bruger prefix notation i Scheme.

Nogle Scheme systemer understøtter 1- og -1, som beskrevet ovenfor. Men det er ikke standard Scheme, så undlad at bruge dem, hvis de er der. Når du flytter dit program til en anden Scheme implementation, duer det måske ikke.

Flere eksempler:

```
(define one-minus (left-section - 1))
```

```
> (one-minus 5) = -4
```

```
(define decrement-by-one (right-section - 1))
```

```
> (decrement-by-one 5) = 4
```

Filtrering.

Filtreringen af listen L med prædikatet P giver den længste delliste af L, hvori alle elementer opfylder P.

```
(define (filter pred)
  (lambda(lst)
    (cond
      ((null? lst) '())
      ((pred (car lst))
       (cons (car lst)
              ((filter pred)
               (cdr lst)))))
      (else ((filter pred)
              (cdr lst)))))

> (let ((lst
        (list 1 2 3 4)))
  (append
   ((filter even) lst)
   ((filter
    (negate even)
    lst)))
  (2 4 1 3))
```

Signatur af filter:

$(A \rightarrow \text{Boolean}) \rightarrow (A^* \rightarrow A^*)$.

```
; negate: (X -> Bool) ->
           (X -> Bool)
(define (negate predicate)
  (lambda(x)
    (not (predicate x))))
```

PS3 - Højereordens funktioner

© Kurt Nørmark, Aalborg Universitet.

10/3/96 s. 9

Noter

For at definitionen foroven af filtrering skal være præcis, skal der gælde at elementerne i en delliste D af en liste L forekommer i samme rækkefølge som elementerne i L.

Man kan naturligvis i filter indføre en tilsvarende optimering som indført for curried mapping (på sliden "Mapping opfattet som et funktionale", se notearket).

Funktionssammensætning og krydsprodukt.

Sammensætning:

```
(define (compose f g)
  (lambda (x)
    (f (g x))))
```

Konstruktion:

```
(define (construct f g)
  (lambda(x)
    (list (f x) (g x))))
```

```
1> (define average (compose / (construct sum length)))
average
```

```
2> (average (list 10 20 30 40))
ERROR: Non-numeric argument to /
1
(100 4)
```

```
3> (average (list 10 20 30 40))
25
```

```
(define (compose f g)
  (lambda (x)
    (apply f (g x))))
```

I Scheme kan vi ikke skelne en *parameterliste* fra en almindelig liste.
I andre funktionsorienterede sprog skelnes mellem *lister* og *tupler* (der bl.a. kan anvendes som parameterlister).

PS3 - Højereordens funktioner

Noter

Signaturen af compose er følgende:

Compose: $(Y \rightarrow Z) \times (X \rightarrow Y) \rightarrow (X \rightarrow Z)$

Compose svarer til funktionssammensætning ("bolle"), som den kendes fra matematik.

Signaturen af construct er følgende:

Construct: $(X \rightarrow Y) \times (X \rightarrow Z) \rightarrow (X \rightarrow (Y \times Z))$

Construct svarer til aggregering af to funktioners resultater på samme input.

I average eksemplet ovenfor ønsker vi at

`(construct sum length)`

bliver selve parameterlisten til divisionsfunktionen `/`. I dette eksempel er vi altså oppe mod forskellen mellem

`(/ '(10 5))`

som fås via den øverste compose (og som giver en fejl) og

`(/ 10 5)`

der fås via den indrammede compose, og som er OK.

Problemet er, at vi ikke kan kende forskel på en tupel og en liste, her som returneret fra `construct`. Man kan sige, at i Scheme er *parameterlister*, i form af tupler, ikke 1. classes objekter.

Reducering.

```
reduce-right (f, (e1, e2, ... ,en)) = f(e1, f(e2, ... f(en-1, en)))  
reduce-left (f, (e1, e2, ... ,en)) = f(f(... f(f(e1, e2), e3).. en-1), en)
```

```
(define (reduce-right f)  
  (lambda (lst)  
    (if (null? (cdr lst))  
        (car lst)  
        (f (car lst)  
            (reduce-right f  
                          (cdr lst)))))))
```

```
(define (reduce-left f)  
  (compose  
    (reduce-right (rev f))  
    reverse))
```

```
(define (reduce-left f)  
  (define (help-reduce f res lst)  
    (if (null? lst)  
        res  
        (help-reduce  
          f  
          (f res (car lst))  
          (cdr lst))))  
  (lambda (lst)  
    (help-reduce f (car lst)  
                  (cdr lst))))
```

Signatur af reduce-left og
reduce-right:

$(T \times T \rightarrow T) \rightarrow T^* \rightarrow T$

Eksempel:

```
1> ((reduce-right -) (list 1 2 3 4 5))  
3  
2> ((reduce-left -) (list 1 2 3 4 5))  
-13
```

PS3 - Højereordens funktioner

Noter

Ideen i reducering er at komponere elementer parvis (enten fra venstre eller fra højre) i en liste af værdier, således at vi gennem gentagen komponering ender op med netop én værdi. Eksempelvis er

$(\text{reduce-left } + \text{ '}(1\ 2\ 3)) = ((1 + 2) + 3) = 6.$

Højre-reduceringen giver det samme resultat, idet + er associativ.

Signaturen af både reduce-left og reduce-right er følgende:

reduce-right: $(T \times T \rightarrow T) \rightarrow T^* \rightarrow T$

I det øverste indrammede skema, hvor vi viser hvordan reduce-funktionerne er definerede, viser vi ikke-curried funktioner, altså funktioner af to parametre. I implementationen nedenfor rammen definerer vi curved funktioner.

I det indrammede skema gør vi os den skjulte antagelse at der mindst er to elementer i den liste af elementer, som vi reducerer. I de viste Scheme funktioner er basistilfældet i rekursionen: "listen er af længden 1", i hvilket tilfælde vi blot returnerer dette ene element. Det er forbundet med begrebsmæssige problemer at give en tom listen til reduceringsfunktionerne (se næste slide og diskussionen der).

Bemærk i reduce-right hvordan vi fanger tilfældet lige inden vi står med en tom liste i lst, og hvordan rekursionsudviklingen standser med at listens to sidste elementer bliver gjort til input til f.

En rekursiv definition af reduce-left i stil med definitionen af reduce-right er vanskelig, idet det er svært og ineffektivt (men ikke umuligt) at arbejde på bagenden af lister i Lisp.

Den halerekursive formulering af help-reduce i reduce-left er dog lige ud ad landevejen (og effektiv).

Bemærk, at der er vist to versioner af reduce-left. Den til venstre er defineret ud fra reduce-right. Den til højre er defineret "direkte" (ved halerekursion).

$((\text{reduce-right } -) (\text{list } 1\ 2\ 3\ 4\ 5)) = (-\ 1\ (-\ 2\ (-\ 3\ (-\ 4\ 5))) = 3.$

$((\text{reduce-left } -) (\text{list } 1\ 2\ 3\ 4\ 5)) = (-\ (-\ (-\ (-\ 1\ 2)\ 3)\ 4)\ 5) = -13.$

Akkumulering.

Reduktion virker ikke på den tomme liste.

Akkumulering defineres til også at virke på den tomme liste. Værdien af akkumulering af den tomme liste angives eksplicit ved en ekstra "startværdi" parameter.

accumulate-right: $(S \times T \rightarrow T) \rightarrow T \rightarrow S^* \rightarrow T$.

Bemærk forskellen i forhold til reduce-right.

```
accumulate-right (f, i, (e1, e2, ... ,en)) = f(e1, f(e2, ... f(en-1, f(en, i))))
```

```
(define (accumulate-right f init)
  (lambda (lst)
    (if (null? lst)
        init
        (f (car lst)
            (accumulate-right f init) (cdr lst)))))
```

PS3 - Højereordens funktioner

Noter

Typen af initial-værdien T behøver ikke at være den samme som elementtypen S af listen, der skal akkumuleres. Det hele kommer til at gå op, hvis akkumuleringsfunktionen har signaturen

$$S \times T \rightarrow T.$$

Signaturen af accumulate-right i uncurried form (altså i forhold til situationen i rammen) er:

$$(S \times T \rightarrow T) \times T \times S^* \rightarrow T.$$

Der akkumuleres en liste med elementtype S . Den binære funktion tager et element af typen S og et element af typen T (inicielt den eksplicit angivne initialværdi i), og returnerer et element af typen T . Hermed ender vi helt naturligt med et element af typen T i hånden.

En praktisk anvendelse af dette:

```
(define (new-append x y)
  ((accumulate-right cons y) x))
```

Signaturen af cons skal her betragtes som

$$\text{cons}: T \times T^* \rightarrow T^*.$$

Herved bliver signaturen af new-append

$$\text{new-append}: T^* \times T^* \rightarrow T^*$$

altså som forventet.

Man kan naturligvis også definere venstre akkumulering; dette overlades dog til læseren.

Bemærk at vi i den indrammede definition af accumulate-right angiver en funktion af hele tre parametre. Dette er usædvanligt i dette kapitel, hvor vi efterhånden har vænnet os til at arbejde med curried funktioner.

Et sammensat eksempel.

Opgaven: Givet to lister L og M af tal. Skriv en funktion, som returnerer listen af alle mulige par (l, m) , hvor l er et tal fra L og m er et tal fra M, og hvor hvor ydermere $l+m$ er et lige tal.

```
(define (pair-list e lst)
  ((map
    (right-section cons e)
    lst ))

(define (pairs-in-list L M)
  ((map
    (right-section pair-list L)
    M ))

(define (all-even-pairs L M)
  ((filter
    (lambda (p)
      (even (+ (car p) (cdr p))))
    ((reduce-left append)
     (pairs-in-list L M )))))
```

1> (pair-list 5 (list 1 2 3 4))
((1 . 5) (2 . 5) (3 . 5) (4 . 5))

2> (pairs-in-list (list 1 2 3 4)
 (list 5 6 7 8))
(((1 . 5) (2 . 5) (3 . 5) (4 . 5))
 ((1 . 6) (2 . 6) (3 . 6) (4 . 6))
 ((1 . 7) (2 . 7) (3 . 7) (4 . 7))
 ((1 . 8) (2 . 8) (3 . 8) (4 . 8)))

3> (all-even-pairs
 (list 1 2 3 4)
 (list 5 6 7 8))
((1 . 5) (3 . 5) (2 . 6) (4 . 6)
 (1 . 7) (3 . 7) (2 . 8) (4 . 8))

PS3 - Højereordens funktioner

Noter

Dette er et typisk eksempel på et problem, som lader sig løse ved brug af mapning, filtrering og reducering. Endvidere illustrerer eksemplet, hvordan øvrige funktionaler er nyttige for at danne de funktioner, som bruges i map, filter og reduce.

Den egentlige årsag til at definere tre funktioner er at kunne illustrere mellemresultater på vej mod den endelige løsning. Det er, som bekendt generelt lettere at forstå en løsning, hvis man definerer og navngiver delløsninger.

(right-section pair-list L) svarer pr. definition til funktionen

(lambda(y) (pair-list y L))

Når denne mappes hen over M får vi *en liste af lister af par*. Den første sådanne liste er alle par dannet ud af L og det første element af M.

Løsning på opgave på slide “Generering af curried funktion”:

Udtrykket

(let ((cm (curry -))) (map cm (list 1 2 3 4 5)))

er en liste af funktioner, der hhv. subtraherer deres respektive parametre fra 1, 2, 3, 4 og 5. Hvis res er den resulterende liste af funktioner får vi følgende:

(map (lambda(x) (x 5)) res) = (-4 -3 -2 -1 0)